<?xml version="1.0" encoding="UTF-8"?>

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
	
	<xsl:include href="../admin.xsl" />
	<xsl:include href="../cp/path.xsl" />
	
	<!-- Запускаем рендер страницы -->
	<xsl:template match="/data">
		<html>
		
		<head>
			<title>
				<xsl:value-of select="$pageTitle" />
			</title>
			<xsl:call-template name="meta-head" />
			<script type="text/javascript" src="{$baseURI}/js/admin/system.js" />
		</head>
		<body>
			<xsl:call-template name="admin-header" />
			
			<div id="blocks">
				<div id="blocks_inner" />
			</div>
			
			<div id="content">
				<div id="content_inner">
					<xsl:apply-templates select="response" />
				</div>
			</div>
			
			<xsl:call-template name="footer" />
		</body>
		</html>
	</xsl:template>
	
	<xsl:template match="response">
		<xsl:param name="cnt" select="count(uris/uri)" />
		
		<div id="path">
			<xsl:apply-templates select="document('xmodule://cp/getPath/0;system;edit_uris')/data/response" mode="cp-path" />
		</div>
		
		<h1>Сокращенные URI</h1>
		
		<xsl:apply-templates select="message" />
		
		<form action="{$baseURI}/admin/system/edit_uris" method="post">
			<div id="uris">
				<xsl:for-each select="uris/uri">
					<p>
						<input type="text" name="uri[{position()}][from]">
							<xsl:attribute name="value">
								<xsl:value-of select="from" />
							</xsl:attribute>
						</input>
						<span> → </span>
						<input type="text" name="uri[{position()}][to]">
							<xsl:attribute name="value">
								<xsl:value-of select="to" />
							</xsl:attribute>
						</input>
					</p>
					
					<xsl:if test="$cnt = position()">
						<p>
							<input type="text" name="uri[{$cnt + 1}][from]" value="" />
							<span> → </span>
							<input type="text" name="uri[{$cnt + 1}][to]" value="" />
							<span> </span>
							<input type="button" value="+" onclick="addURI({$cnt + 2})" />
						</p>
					</xsl:if>
				</xsl:for-each>
			</div>
			
			<p>
				<input type="submit" name="submit" value="Сохранить изменения" />
			</p>
			<p>В левом столбце записывается часть URI (начиная с начала), которая будет подменена (незаметно для пользователя) на значение в правом столбце. Например:<br /><i>www.example.com/<b>user</b>/admin</i> → <i>www.example.com/<b>users/profile</b>/admin</i><br /><i>www.example.com/<b>concats</b></i> → <i>www.example.com/<b>page/show/concats</b></i></p>
			<p><b>Внимание!</b> Изменение сокращенных URI может привести к неработоспособности некоторых ссылок.</p>
		</form>
	</xsl:template>
	
</xsl:stylesheet>